

Hewlett-Packard

HP-19C/HP-29C SOLUTIONS

MECHANICAL ENGINEERING

INTRODUCTION

This HP-19C/HP-29C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become an expert on your HP calculator.

You will find general information on how to key in and run programs under "A Word about Program Usage" in the Applications book you received with your calculator.

We hope that this Solutions book will be a valuable tool in your work and would appreciate your comments about it.

TABLE OF CONTENTS

1. RPM/TORQUE POWER 1

   This program provides interchangeable solutions for RPM/Torque, and power.

2. CRITICAL SHAFT SPEED 4

   This program finds the fundamental critical speed of a rotating shaft.

3. LINEAR PROGRESSION OF A SLIDER CRANK 8

   This program calculates the displacement, velocity, and acceleration of the slide in a slider crank mechanism.

4. SPUR GEAR REDUCTION DRIVE 12

   This program provides interchangeable solutions for reduction, distance between centers, diametral pitch, and number of pinion teeth. The program also outputs values for the pitch diameters of the pinion and the gear, and the number of gear teeth.

5. BELT LENGTH 16

   This program computes the belt length around an arbitrary set of pulleys.

6. REVERSIBLE POLYTROPIC PROCESS FOR AN IDEAL GAS 20

   This program provides interchangeable solutions between pressure ratio, volume ratio, temperature ratio, and density ratio.

7. ISENTROPIC FLOW FOR AN IDEAL GAS 23

   This program replaces isentropic flow tables for a specified specific heat ratio.

8. HEAT TRANSFER THROUGH COMPOSITE CYLINDERS AND WALLS 27

   This program calculates the overall heat transfer coefficient from individual section conductances and surface coefficients.

9. BLACKBODY THERMAL RADIATION. 31

   This program calculates the wavelength of maximum emissive power for a given temperature (or vice versa), and the emissive power (total, from λ, to λ_2, or at λ_1).

10. CONSERVATION OF ENERGY 36

    This program converts kinetic energy, potential energy, and pressure-volume work to energy, sums all the energy contributions, and converts the total to an equivalent velocity, height, pressure, or energy per unit mass.

RPM/TORQUE/POWER

This program provides an interchangeable solution for RPM, torque, and power in both Systeme International (metric) and English units.

	SI	English
RPM	RPM	RPM
Torque	nt-m	ft-lb
Power	watts	hp

EQUATIONS:

RPM x Torque = Power

1 hp = 745.7 watts

1 ft-lb = 1.356 joules

1 RPM = $\pi/30$ radians/sec

1 hp = 550 $\frac{\text{ft-lb}}{\text{sec}}$

EXAMPLE 1:

Calculate the torque from an engine developing 11 hp at 6500 RPM. Find the SI equivalent.

EXAMPLE 2:

A generator is turning at 1600 RPM with a torque of 20 nt-m. If it is 90% efficient, what is the power input in both systems?

SOLUTIONS:

(1)

```
          GSB4
6500.00   ENT↑
   0.00   ENT↑
  11.00   GSB5
   8.89   ***    Torque, ft-lb
          R/S
  12.05   ***    Torque, nt-m
```

(2)

```
          GSB3
1600.00   ENT↑
  20.00   ENT↑
   0.90   ÷
   0.00   GSB5
3723.37   ***    Power, watts
          R/S
   4.99   ***    Power, hp
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Choose sytem of units:				
	Metric		GSB	3	1.36
	or				
	English		GSB	4	5251.41
3.	Enter variables (the unknown quantity, x,				
	must be input as zero) and compute x.				
	RPM	RPM	ENT↑		
	Torque	Torque	ENT↑		
	Power	Power	GSB	5	x
4.	(Optional) to convert torque or power to				
	other system		R/S		x converted

Program Listings

```
01 *LBL3
02    3        Set up for metric
03    0          units
04    Pi
05    ÷
06  STO7
07    7
08    4
09    5
10    .
11    7
12  STO5
13    1
14    .
15    3
16    5
17    6
18  STO6
19   RTN
20 *LBL4       Set up for English
21  GSB3         units
22   1/X
23  STO6
24   X⇄Y
25   1/X
26  STO5
27    ÷
28    x
29  STO7
30   RTN
31 *LBL5
32    4
33  STO0
34   R↓
35 *LBL8
36  STOi       Store variables
37   R↓
38   DSZ
39  GTO8
40 *LBL9
41  X=0?       Determine quantity
42  GTOi         to calculate
43   ISZ
44   R↓
45  GTO9
46 *LBL2
47  RCL4
48  RCL3
49    ÷
50  RCL7
51    x
52   R/S       ** RPM
53 *LBL1
54  RCL4
55  RCL2
56    ÷
57  RCL7
58    x
59   R/S       *** Torque
60  RCL6
61    ÷
62   R/S       ** Torque converted
63 *LBL0
64  RCL2
65  RCL3
66    x
67  RCL7
68    ÷
69   R/S       *** Power
70  RCL5
71    ÷
72   R/S       ** Power converted
```

** "Printx" may be inserted before "R/S".

*** "Printx" may replace "R/S".

REGISTERS

0 i	1 Used	2 RPM	3 Torque	4 Power	5 Used
6 Used	7 Used	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

CRITICAL SHAFT SPEED

Suppose a rotating shaft is simply supported at both ends and has a series of n weights, $W_1, \dots, W_n$, attached. Then there are critical speeds at which the shaft will become dynamically unstable. This program finds the fundamental critical speed from the formula

$$f = \frac{1}{2\pi}\sqrt{\frac{g\sum_{i=1}^{n} W_i y_i}{\sum_{i=1}^{n} W_i y_i^2}} \quad \text{cycles/sec}$$

where

g = Acceleration due to gravity

y_i = Static deflection of weight W_i

The program is set up to accept the static deflections y_i as inputs. If the static deflections are not known, it calculates y_{ij}, the static deflection of weight i due to W_j. Then the total deflection of weight i is the sum of the deflections from all the W_j's. That is,

$$y_i = \sum_{j=1}^{n} y_{ij}.$$

The individual y_{ij}'s are added to provide the y_i's which the program accepts as inputs. The y_{ij}'s are calculated as follows:

If $x_i < x_j$

$$y_{ij} = \frac{W_j(\ell - x_j)x_i}{6\ell EI}\left[\ell^2 - (\ell - x_j)^2 - x_i^2\right]$$

$$= \frac{W_j(\ell - x_j)x_i}{6\ell EI}\left[2\ell x_j - x_j^2 - x_i^2\right]$$

If $x_i \geq x_j$

$$y_{ij} = \frac{W_j x_j(\ell - x_i)}{6\ell EI}\left[\ell^2 - x_j^2 - (\ell - x_i)^2\right]$$

$$= \frac{W_j x_j(\ell - x_i)}{6\ell EI}\left[2\ell x_i - x_j^2 - x_i^2\right]$$

where

x_i, x_j = Distance of weights i,j from end of shaft

E = Modulus of elasticity

I = Moment of inertia = $\frac{\pi d^4}{64}$

ℓ = Length of shaft



Any consistent set of units may be used. The acceleration due to gravity, g, will of course change from one set of units to another. Some useful values are listed below:

g = 32.1740 ft/sec^2
= 386.088 in/sec^2
= 9.80665 m/sec^2
= 980.665 cm/sec^2

REFERENCE: Design of Machine Elements, M.F. Spotts, Prentice-Hall, 1971.

EXAMPLE: A 2 inch diameter steel shaft of total length 40 inches has a fly-wheel and a gear located respectively 15 and 25 inches from the end. The flywheel weights 60 pounds and the gear 45 pounds. Assume the modulus of elasticity of the steel is 30 x 10^6 psi. Find the fundamental critical speed of the shaft.



SOLUTION:

```
    2.00 ENT↑
  30.+06 ENT↑
    2.00 GSB6    d
   40.00 GSB1
   60.00 ENT↑
   15.00 GSB2
   45.00 ENT↑
   25.00 GSB3
   45.00 ENT↑
   25.00 GSB2
   60.00 ENT↑
   15.00 GSB3
 386.088 GSB5
   44.15  ***    f,cycles/sec
   60.00   x
 2648.85  ***    f, RPM
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	If y_i are known, go to step 10				
3.	If the y_i are not known, input	n	↑		
	Modulus of elasticity	E	↑		E
	Moment of inertia**	I	↑		I
	Length of shaft	ℓ	GSB	1	n
4.	Repeat steps 5-7 for i = 1,...,n				
5.	Input W_i	W_i	↑		W_i
	x_i	x_i	GSB	2	Index
6.	Repeat step 7 for all $j \neq i$				
7.	Input W_j	W_j	↑		W_j
	x_j where $j \neq i$	x_j	GSB	3	Index or $W_i y_i^2$
8.	Input acceleration of gravity and calculate				
	critical speed.	g	GSB	5	f(cycles/sec)
9.	For a new case, go to step 2				
10.	If the y_i are known, input length of shaft	ℓ	GSB	1	
11.	Repeat step 12 for i = 1,..., n				
12.	Input W_i	W_i	ENT↑		W_i
	y_i	y_i	GSB	4	$W_i y_i^2$
13.	Input acceleration of gravity and compute				
	critical speed	g	GSB	5	f(cycles/sec)
14.	For a new case go to step 2				
	**If I is not known, it may be calculated	d	GSB	6	I
	from the diameter (solid cylinderical				
	shaft only).				

Program Listings

01	*LBL1	
02	CLRG	
03	STO8	ℓ
04	x	
05	x	
06	6	
07	x	
08	STO9	GEIℓ
09	R↓	
10	ST.0	n
11	R/S	
12	*LBL2	
13	STO5	x_i
14	X⇄Y	
15	STO4	W_i
16	0	
17	STO3	
18	R↓	
19	X⇄Y	x_iW_i
20	*LBL3	x_jW_j
21	ISZ	
22	RCL5	x_i
23	X>Y?	
24	X⇄Y	
25	STO6	
26	X²	
27	X⇄Y	
28	STO7	
29	X²	
30	+	
31	RCL8	
32	RCL7	
33	x	
34	2	
35	x	
36	-	
37	x	
38	RCL6	
39	x	
40	RCL7	
41	RCL8	
42	-	
43	x	
44	RCL9	
45	÷	y_{ij}
46	ST+3	
47	RC.0	
48	RCL0	
49	X≠Y?	Index ≠ n?
50	R/S	
51	0	
52	STO0	
53	RCL4	W_i
54	RCL3	y_i
55	*LBL4	
56	x	y_iW_i
57	ST+1	
58	LSTX	
59	x	
60	ST+2	$y_i^2W_i$
61	R/S	
62	*LBL5	
63	RCL1	
64	x	
65	RCL2	
66	÷	
67	√X	
68	Pi	
69	÷	
70	2	
71	÷	
72	R/S	**f
73	*LBL6	d
74	4	
75	Y^X	
76	Pi	
77	x	
78	6	
79	4	
80	÷	
81	R/S	I

**"Printx" may be inserted before "R/S".

REGISTERS

0 Index	1 ΣW_iy_i	2 $\Sigma W_iy_i^2$	3 $\Sigma y_{ij}=y_i$	4 W_i	5 x_i
6 Min(x_i,x_j)	7 Max(x_i,x_j)	8 ℓ	9 GEIℓ	.0 n	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

LINEAR PROGRESSION OF SLIDER CRANK



This program calculates the displacement, velocity, and acceleration of the slider in a slider crank mechanism, (e.g. the piston wrist-pin in an internal combustion engine) given crank radius, connecting rod length, slider offset, crankshaft speed, and crank position. The maximum and minimum displacements and the stroke are also calculated.

N = Crankshaft speed, RPM

E = Slider offset

L = Connecting rod length

R = Crank radius

ω = Crank angular velocity, radians/sec

θ = Crank angle

x = Slider displacement

x_{max} = Maximum slider displacement

x_{min} = Minimum slider displacement

Δx = Stroke

v = Slider velocity

a = Slider acceleration

ϕ = Connecting rod angle

EQUATIONS:

$$\omega = \frac{\pi N}{30}$$

$$x = R\cos\theta + L\cos\phi$$

$$x_{max} = (R + L)\cos\left[\sin^{-1}\left(\frac{E}{R + L}\right)\right]$$

$$x_{min} = (L - R)\cos\left[\sin^{-1}\left(\frac{E}{L - R}\right)\right]$$

$$\Delta x = x_{max} - x_{min}$$

$$\phi = \sin^{-1}\left(\frac{E + R\sin\theta}{L}\right)$$

$$v = \frac{dx}{dt} = R\omega\left(\frac{-\sin(\theta+\phi)}{\cos\phi}\right)$$

$$a = \frac{d^2x}{dt^2} = R\omega^2\left(\frac{-\cos(\theta+\phi)}{\cos\phi} - \frac{R\cos^2\theta}{L\cos^3\phi}\right)$$

REFERENCES:

Mechanical Design and Systems Handbook, H.A. Rothbart, McGraw-Hill, 1964.

Kinematics, V.M. Faires, McGraw-Hill, 1959.

EXAMPLE:

Find the displacement, velocity and acceleration of the wrist-pin in the slider of a slider crank mechanism having a crank radius of 2.0 inches and connecting rod length of 7.0 inches, turning at 4800 RPM. Calculate values for

$\theta = 0°, 15°, 45°, 90°, 135°, 180°, 225°$

Assume the slider crank mechanism is in-line (E=0). Also find the maximum and minimum displacements and the stroke.

SOLUTION:

4800.00	ENT↑	
0.00	ENT↑	
7.00	ENT↑	
2.00	GSB1	
502.65	***	ω rad./sec
0.00	GSB2	
9.00	***	x in.
	R/S	
9.00	***	x_{max}, in.
	R/S	
5.00	***	x_{min}, in.
	-	
4.00	***	Δx, in.
	GSB3	
0.00	***	v, in./sec.
	GSB4	
-649701.96	***	a, in./sec.^{2}
15.00	GSB2	
8.91	***	x, in.
	GSB3	
-332.20	***	v, in./sec.
	GSB4	
-614226.44	***	a, in./sec.^{2}

.

.

.

225.00	GSB2	
5.44	***	x, in.
	GSB3	
564.22	***	v, in./sec.
	GSB4	
354181.29	***	a, in./sec.^{2}

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Input the slider crank parameters	N	ENT↑		N
		E	ENT↑		E
		L	ENT↑		L
	and display the crank angular velocity	R	GSB	1	ω
3.	Input the crank angle and compute the				
	slider displacement	θ	GSB	2	x
	Optional: calculate the maximum displace-				
	ment		R/S		x_{max}
	Optional: calculate the minimum				
	displacement		R/S		x_{min}
	Optional: calculate the stroke		-		Δx
4.	Calculate the slider velocity		GSB	3	v
5.	Calculate the slider acceleration		GSB	4	a
6.	Repeat step 3 for all desired values of θ.				
	Steps 4 and 5 may be executed (in order)				
	as required.				
7.	For a new case, go to step 2.				

Program Listings

01 *LBL1	
02 STO1	R
03 R↓	
04 STO2	L
05 R↓	
06 STO3	E
07 R↓	
08 Pi	
09 x	
10 3	
11 0	
12 ÷	
13 STO4	**ω
14 R/S	θ
15 *LBL2	
16 STO5	
17 SIN	
18 RCL1	
19 x	
20 RCL3	
21 +	
22 RCL2	
23 ÷	
24 SIN⁻¹	
25 STO6	ϕ
26 COS	
27 STO7	cosϕ
28 RCL2	
29 x	
30 RCL5	
31 COS	
32 RCL1	
33 x	
34 +	
35 R/S	*** x
36 RCL3	
37 RCL2	
38 RCL1	
39 *LBL0	
40 +	
41 STO8	R + L or L - R
42 ÷	
43 SIN⁻¹	
44 COS	
45 RCL8	
46 x	
47 R/S	** x_{max} or x_{min}

48 RCL3	
49 RCL2	
50 RCL1	
51 CHS	
52 GTO0	Calculate x_{min}
53 *LBL3	
54 RCL6	
55 RCL5	
56 +	
57 RCL1	
58 CHS	-R θ+ϕ
59 →R	
60 STO8	
61 CLX	
62 RCL4	
63 x	
64 RCL7	
65 ÷	
66 R/S	**v
67 *LBL4	-R cos (θ+ϕ)
68 RCL8	
69 RCL5	
70 COS	
71 RCL7	
72 ÷	
73 RCL1	
74 x	
75 X²	
76 RCL2	
77 ÷	
78 -	
79 RCL4	
80 X²	
81 x	
82 RCL7	
83 ÷	
84 R/S	**a

** "Printx" may be inserted before "R/S".

*** "Printx" may replace "R/S".

REGISTERS

0	1 R	2 L	3 E	4 ω	5 θ
6 ϕ	7 cos ϕ	8 Used	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

SPUR GEAR REDUCTION DRIVE



For a spur gear meshing with a pinion, this program performs an interchangeable solution among the variables reduction (f), distance between the centers (C.D.), diametral pitch (P), and number of pinion teeth (N_p). Once these four basic variables have been determined, the program will also output values for the pitch diameters of the pinion and the gear (D_p and D_g) and the number of gear teeth (N_g).

The basic formula used in all solutions is:

$$f + 1 = \frac{2P \times C.D.}{N_p} \qquad (1)$$

The calculations for f,P, and C.D. are straightforward. The solution for N_p is more complicated since it must be an integer. Because of this constraint, there may not be a gear-pinion combination that will give exactly the desired reduction. In this case, the program finds the closest integer value for N_p by the formula

$$N_p = \text{INT}\left(\frac{2P \times C.D.}{f + 1} + 0.5\right)$$

where INT (x) = the integer portion of x.

Then a new value for the reduction, f', is found by substituting this N_p into equation (1) above. The next step is to compute the number of gear teeth (also an integer) by

$$N_g = \text{INT}\ (f'N_p + 0.5).$$

Finally the true value of the reduction is found by

$$f = \frac{N_g}{N_p}$$

This modified value for f is stored in R_1 and may be recalled by the user if desired.

<u>REMARKS</u>:

The program assumes that the reduction will be expressed as a decimal number greater than 1. For instance, a reduction of 9:2 should be input as $\frac{9}{2}$, or 4.5. If f<1, the program will still work but the pinion values and gear values will be reversed.

<u>REFERENCE:</u>

Design of Machine Elements, M.F. Spotts, Prentice-Hall, 1971.

<u>EXAMPLE</u>:

A spur gear reduction mechanism is to be designed to reduce a rotation from 1800 RPM to 650 PRM. The distance between the centers of the gear and pinion is constrained to be 9 inches. If the designer wishes to use teeth of diametral pitch 9, how many teeth should be on the pinion? On the gear (38,106) What will the pitch diameters of the gears be? (4.75 inches, 13.25 inches) What is the actual reduction in speed? (2.79)

SOLUTION:

```
1800.00 ENT↑
 650.00   ÷
   2.77  ***   f design
   9.00 ENT↑
   8.00 ENT↑
   0.00 GSB1
  38.00  ***   Np
        GSB2
   4.75  ***   Dp
         R/S
 106.00  ***   Ng
         R/S
  13.25  ***   Dg
        RCL1
   2.79  ***   f
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Enter variables (the unknown quantity, x,				
	must be input as zero) and compute x.				
	Reduction	f	ENT↑		
	Center distance	C.D.	ENT↑		
	Diametral pitch	P	ENT↑		
	Number of pinion teeth	N_p	GSB	1	x
3.	Display the following variables:				
	Pitch diameter of pinion		GSB	2	D_p
	Number of gear teeth		R/S		N_g
	Pitch diameter of gear		R/S		D_g
4.	To display any of the basic variables:				
	Reduction		RCL	1	f
	Center distance		RCL	2	
			2	÷	C.D.
	Diametral pitch		RCL	3	P
	Number of pinion teeth		RCL	4	N_p
5.	To change any inputs, go to step 2				

Program Listings

01	*LBL1	
02	STO4	
03	R↓	
04	STO3	
05	R↓	
06	STO2	
07	ST+2	
08	R↓	
09	STO1	
10	X=0?	f=0?
11	GTO0	
12	R↓	
13	X=0?	N_p=0?
14	GTO9	
15	R↓	
16	X=0?	P=0?
17	GTO8	
18	GTO6	CD=0
19	*LBL0	
20	RCL2	
21	RCL3	
22	x	
23	RCL4	
24	÷	
25	1	
26	-	
27	STO1	
28	R/S	***f
29	*LBL9	
30	RCL2	
31	RCL3	
32	x	
33	RCL1	
34	1	
35	+	
36	÷	
37	.	
38	5	
39	+	
40	INT	
41	STO4	N_p
42	CHS	
43	RCL2	
44	RCL3	
45	x	
46	+	f'N_p
47	.	
48	5	
49	+	
50	INT	N_g
51	RCL4	
52	÷	
53	STO1	f
54	RCL4	
55	R/S	**N_p
56	*LBL8	
57	RCL4	
58	RCL1	
59	1	
60	+	
61	x	
62	RCL2	
63	÷	
64	STO3	
65	R/S	**P
66	*LBL6	
67	RCL1	
68	1	
69	+	
70	RCL4	
71	x	
72	RCL3	
73	÷	
74	STO2	
75	2	
76	÷	
77	R/S	**C.D.
78	*LBL2	
79	RCL4	
80	RCL3	
81	÷	
82	STO5	
83	R/S	***D_p
84	RCL4	
85	RCL1	
86	x	
87	STO6	
88	R/S	***N_g
89	RCL5	
90	RCL1	
91	x	
92	STO7	
93	R/S	**Dg

** "Printx" may be inserted before "R/S".
***"Printx" may replace "R/S".

REGISTERS

0	1 f	2 C.D.	3 P	4 N_p	5 D_p
6 N_g	7 Dg	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

BELT LENGTH



$$\theta_{12} = \tan^{-1}\left(\frac{y_2-y_1}{x_2-x_1}\right)$$

$$\theta_{23} = \tan^{-1}\left(\frac{y_3-y_2}{x_3-x_2}\right)$$

This program generates accurate results for any convex polygon, i.e., a line between any two points within the region bounded by the center-to-center line segments is entirely contained within the region.



This program calculates the belt length around an arbitrary set of pulleys. It may also be used to calculate the total length between any connected set of coordinates. The program assumes the coordinates of the first pulley to be (0,0).

(x_i, y_i, R_i) = x,y coordinates and radius of pully i

R_0 = Radius of first pulley

C.D. = Center to center distance of consecutive pulleys

L = Total length of belt

<u>EQUATIONS</u>:

$$L_{12} = \sqrt{C.D._{12}{}^2 - (R_2-R_1)^2}$$

$$\text{Arc Length}_2 = R_2(\pi-\alpha-\beta-\gamma_2)$$

$$\alpha = \tan^{-1}\left(\frac{R_1-R_2}{L_{12}}\right)$$

$$\beta = \tan^{-1}\left(\frac{R_3-R_2}{L_{23}}\right)$$

$$\gamma = \theta_{12}-\theta_{23}$$

In some cases, there are two physically possible directions for the belt to take:



The program chooses the upper side if the middle pulley center lies above the line connecting the previous and following pulleys:

Case 1



The program chooses the lower side if the middle pulley center lies below the line connecting the previous and following pulleys:

Case 2



The program generates inaccurate answers in the second case. Note the figure bounded by the center-to-center line segments for the second case is not convex.

REMARKS:

The calculator is set and left in radians mode.

EXAMPLE 1:

Assume three pulleys are positioned as shown below with the following coordinates and radii:

Pulley 1 (0,0,4 inches)

Pulley 2 (-8,15,1.5 inches)

Pulley 3 (9,16,1 inches)

Find the belt length around the three pulleys. (66.53 inches)



EXAMPLE 2:

Find the length of line connecting the points (0,0), (1.5,7), (3.2,-6), (0,0.5), (0,0). (L = 28.01). Let the radius of each "pulley" be 0.

SOLUTION:

1.

```
 4.00 GSB1
-8.00 ENT↑
15.00 ENT↑
 1.50 GSB2
 9.00 ENT↑
16.00 ENT↑
 1.00 GSB2
 0.00 ENT↑
 0.00 ENT↑
 4.00 GSB2
      GSB3
66.53  ***    L
```

2.

```
 0.00 GSB1
 1.50 ENT↑
 7.00 ENT↑
 0.00 GSB2
 3.20 ENT↑
-6.00 ENT↑
 0.00 GSB2
 0.00 ENT↑
 0.50 ENT↑
 0.00 GSB2
 0.00 ENT↑
 0.00 ENT↑
 0.00 GSB2
      GSB3
28.01  ***    L
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Enter the initial pulley radius	R_0	GSB	1	1.00
3.	Enter the next pulley coordinates and	x_i	ENT↑		
	radius	y_i	ENT↑		
		R_i	GSB	2	R_i
4.	Repeat step 3 for all pulleys.				
	(the last pulley to be entered is the				
	origin pulley).				
5.	Compute belt length		GSB	3	L

Program Listings

```
01 *LBL1
02   RAD
03  CLRG
04  STO1        R0
05     1
06  STO0        Set flag
07   R/S
08 *LBL2
09  STO4
10   CLX
11  RCL3
12   X⇄Y
13  STO3
14     -
15   X⇄Y
16  RCL2
17   X⇄Y
18  STO2
19     -
20    →P
21    X²        C.D.θ
22   X⇄Y
23  X>0?
24  GTO0
25     2
26    Pi
27     x
28     +
29 *LBL0
30   DSZ        Test flag
31  GTO9
32  STO5
33  STO6
34 *LBL9
35  RCL6
36   X⇄Y
37  STO6
38  GSB8
39   X⇄Y        C.D.²
40  RCL1
41  RCL4
42     -
43    X²
44     -
45    √X
46  ST+8        L i-1,i
47  RCL1
48  RCL4
49     -
50   X⇄Y
51     ÷
52  TAN⁻¹       -β
53  STO7        α
54     +
55  RCL1
56     x        Arc length i
57  ST+8
58  RCL4
59  STO1
60   R/S
61 *LBL8
62     -        γ
63     1
64    →R        Resolves to less
65    →P        than 2 radians
66     x
67   ABS
68  RCL7        α
69     -
70   RTN
71 *LBL3
72  RCL6
73  RCL5
74  GSB8
75  RCL1
76     x
77  RCL8
78     +
79   DEG        Restore "normal"mode
80   R/S        ** L
```

** "Printx" may be inserted before "R/S".

REGISTERS

0 Flag	1 R_{i-1}	2 x_i	3 y_i	4 R_i	5 θ_o
6 θ_i	7 α	8 Σ length	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

REVERSIBLE POLYTROPIC PROCESS FOR AN IDEAL GAS

This program may be used to solve interchangeably between pressure ratio, volume ratio, temperature ratio, and density ratio for polytropic processes involving ideal gases. Polytropic processes are defined by the relation

$$PV^n = C$$

which is shown graphically in Figure 1.



Figure 1.

Isentropic processes are special cases of polytropic processes. For isentropic processes, k, the specific heat ratio, is equal to n.

EQUATIONS:

$$\frac{P_2}{P_1} = \left(\frac{V_2}{V_1}\right)^{-n} = \left(\frac{T_2}{T_1}\right)^{\frac{n}{n-1}} = \left(\frac{\rho_2}{\rho_1}\right)^{n}$$

where

P_2/P_1 is the final pressure divided by the inital pressure;

V_2/V_1 is the final volume divided by the inital volume;

T_2/T_1 is the final temperature divided by the initial temperature;

ρ_2/ρ_1 is the final density divided by the initial density.

EXAMPLE: A compressor has a compression ratio of 8.5 (V_1/V_2). The polytropic constant is 1.43. If inlet air is at 300K, what is outlet temperature? What is the pressure in atmospheres if the inlet pressure is one atmosphere?

SOLUTION:

```
  1.43 GSB1
  8.50  1/X
       GSB3
       PRST   (for 29C manually
               review the stack)
  8.50    T   ρ2/ρ1
  0.12    Z   V2/V1
  2.51    Y   T2/T1
 21.33    X   P2/P1

       RCL8
300.00    x
752.96  ***   Outlet temp.(K)
       RCL5
  1.00    x
 21.33  ***   Pressure (atm)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Enter polytropic constant	n	GSB	1	n
3.	Enter one of the following:				
	Pressure ratio	P_2/P_1	GSB	2	P_2/P_1
	Volume ratio	V_2/V_1	GSB	3	P_2/P_1
	Temperature ratio	T_2/T_1	GSB	4	P_2/P_1
	Density ratio	ρ_2/ρ_1	GSB	5	ρ_2/ρ_1
4.	The four parameters are now in the stack				
	or may be recalled from storage:				
	Pressure ratio (x-register)		RCL	5	P_2/P_1
	Temperature ratio (y-register)		RCL	8	T_2/T_1
	Volume ratio (z-register)		RCL	7	V_2/V_1
	Denisty ratio (t-register)		RCL	6	ρ_2/ρ_1

Program Listings

01 *LBL1	n
02 STO2	
03 1	
04 -	
05 STO3	
06 RCL2	
07 ST÷3	
08 R/S	
09 *LBL2	P_2/P_1
10 1	
11 GTO0	
12 *LBL3	V_2/V_1
13 RCL2	
14 CHS	
15 GTO0	
16 *LBL4	T_2/T_1
17 RCL3	
18 1/X	
19 GTO0	
20 *LBL5	P_2/P_1
21 RCL2	
22 *LBL0	
23 Y^x	
24 STO5	P_2/P_1
25 RCL2	
26 1/X	
27 Y^x	ρ_2/ρ_1
28 STO6	
29 ENT↑	
30 1/X	V_2/V_1
31 STO7	
32 RCL5	
33 RCL3	
34 Y^x	T_2/T_1
35 STO8	
36 RCL5	
37 R/S	*** $\frac{P_2}{P_1}, \frac{T_2}{T_1}, \frac{V_2}{V_1}, \frac{\rho_2}{\rho_1}$

*** "Print Stack" may be inserted before "R/S".

REGISTERS

0	1	2 n	3 $(n-1)/n$	4	5 P_2/P_1
6 ρ_2/ρ_1	7 V_2/V_1	8 T_2/T_1	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

ISENTROPIC FLOW FOR IDEAL GASES

This program can be used to replace flow tables for a specified specific heat ratio, k.

EQUATIONS:

$$A/A^* = \frac{1}{M}\left[\left(\frac{2}{k+1}\right)\left(1 + \frac{k-1}{2}M^2\right)\right]^{\frac{k+1}{2(k-1)}}$$

$$T/T_0 = \frac{2}{2 + (k-1)\,M^2}$$

$$P/P_0 = (T/T_0)^{k/(k-1)}$$

$$\rho/\rho_0 = (T/T_0)^{1/(k-1)}$$

where

- M the mach number;
- T/T_0 the ratio of flow temperature T to static or zero velocity temperature T_0;
- P/P_0 the ratio of flow pressure P to static pressure P_0;
- ρ/ρ_0 the ratio of flow density ρ to static density ρ_0;
- A/A^*_{sub}, and A/A^*_{sup} are the ratios of flow area A to the throat area A* in converging-diverging passages. A/A^*_{sub} refers to subsonic flow while A/A^*_{sup} refers to supersonic flow.

M^2 is determined using Newton's method. The initial guess used is as follows with a positive exponent for supersonic flow:

$$M_0{}^2 = (\sqrt{\text{Frac}(A/A^*)} + A/A^*)^{\pm 3}$$

REMARKS:

After an input of A/A* the program begins to iterate to find M^2 for future use. This iteration will normally take less than one minute, but may take longer on occasion and for extreme values of k (1.4 is optimum) may fail to converge at all.

A/A* values of 1.00 are illegal inputs. M = 1 in this case.

EXAMPLE 1:

A pilot is flying at mach 0.93 and reads an air temperature of 15 degrees Celsius (288 K) on a thermometer that reads stagnation temperature T_0. What is the true temperature assuming that k = 1.38?

If the pilot reads a stagnation pressure P_0 of 28 inches of mercury, what is the true air pressure?

EXAMPLE 2:

A converging, diverging passage has supersonic flow in the diverging section. At an area ratio A/A* of 1.60, what are the isentropic flow ratios for temperature, pressure and density? What is the mach number? k = 1.74.

SOLUTION:

1.

1.38	GSB1	
0.93	X^2	
	STO1	M^2
	GSB3	
1.00	***	A/A*
	GSB9	
	RCL8	T/T_0
0.86	***	
288.00	x	
247.35	***	T(°K)
	RCL5	
0.58	***	P/P_0
28.00	x	
16.11	***	P (in. Hg)

2.

1.74	GSB1	
1.60	GSB2	
2.11	***	M
	R↓	
0.27	***	ρ/ρ_0
	R↓	
0.10	***	P/P_0
	R↓	
0.38	***	T/T_0

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Enter specific heat ratio of gas	k	GSB	1	k + 1
3.	If M is known, then go to step 7.				
4.	Enter area ratio. Use positive values	± A/A*	GSB	2	M
	for supersonic area ratios and negative				
	values for subsonic area ratios				
5.	All four parameters are now in the stack				
	or may be recalled from storage:				
	Mach number (x-register)		RCL	1	
			f	$\sqrt{x}$	M
	Density ratio (y-register)		RCL	6	ρ/ρ_o
	Pressure ratio (z-register)		RCL	5	P/P_o
	Temperature ratio (t-register)		RCL	8	T/T_o
6.	For a new case, go to step 2				
7.	Enter the Mach number	M	g	x^2	
			STO	1	
8.	Calculate the area ratio		GSB	3	A/A*
			GSB	9	M
9.	To determine the remaining parameters go				
	to step 5.				

Program Listings

Step		Comment	Step		Comment
01	*LBL1	k	48	ABS	
02	STO4		49	EEX	
03	STO7		50	CHS	
04	1		51	4	
05	ST+7		52	X≤Y?	Change≥.01%?
06	-		53	GTO0	
07	STO3		54	GTO9	
08	ST÷7		55	*LBL3	
09	2		56	RCL1	
10	ST÷7		57	RCL3	
11	+		58	x	
12	STO2		59	2	
13	R/S		60	+	
14	*LBL2	± A/A*	61	RCL2	[ ]
15	ENT↑		62	÷	
16	ABS		63	STO8	
17	÷		64	RCL7	
18	LSTX	Λ/A* ± 1	65	Y^x	
19	STO6		66	RCL1	
20	ENT↑		67	√X	
21	FRC		68	÷	** A/A*
22	√X		69	RTN	
23	+		70	*LBL9	Calculate T/T_o
24	X⇄Y		71	2	
25	3		72	RCL1	
26	x	±3	73	RCL3	
27	Y^x		74	x	
28	STO1	M_o^2	75	2	
29	*LBL0		76	+	
30	RCL6		77	÷	
31	GSB3	A/A*	78	STO8	
32	÷		79	RCL8	
33	1		80	RCL4	
34	-		81	RCL3	
35	.		82	÷	
36	5		83	Y^x	P/P_o
37	RCL8		84	STO5	
38	÷		85	RCL5	
39	.		86	RCL4	
40	5		87	1/X	
41	RCL1		88	Y^x	
42	÷		89	STO6	ρ/ρ_o
43	-		90	RCL1	
44	÷	ΔM^2	91	√X	M
45	ST+1		92	R/S	*** $M, \frac{\rho}{\rho_o}, \frac{P}{P_o}, \frac{T}{T_o}$
46	RCL1				
47	÷				

** "Printx" may be inserted before "RTN".
*** "Print Stack" may be inserted before "R/S".

REGISTERS

0	1 M^2	2 k + 1	3 k - 1	4 k	5 P/P_o
6 ρ/ρ_o	7 $(k+1)/(k-1)^2$	8 Used, T/T_o	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

HEAT TRANSFER THROUGH COMPOSITE CYLINDERS AND WALLS



Figure 1.—Composite tube

This program can be used to calculate the overall heat transfer coefficient for composite tubes and walls from individual section conductances and surface coefficients.

Equations:

The overall heat transfer coefficient U is defined by:

$$q/L = U\,\Delta T$$

or

$$q/A = U\,\Delta T$$

where ΔT is the total temperature difference ($T_2 - T_1$), q/L is the heat transfer per unit length of pipe, and q/A is the heat transfer per unit area of wall.

For cylinders

$$U = \frac{2\pi}{\dfrac{2}{h_1 D_1} + \dfrac{\ln D_2/D_1}{k_1} + \dfrac{\ln D_3/D_2}{k_2} + \ldots + \dfrac{2}{h_n D_n}}$$

For walls

$$U = \frac{1}{\dfrac{1}{h_1} + \dfrac{x_1}{k_1} + \dfrac{x_2}{k_2} + \ldots + \dfrac{1}{h_n}}$$

where
- h is the convective surface coefficient:
- D_n is the outside diameter of the annulus:
- k is the conductive coefficient;
- x is the thickness of a wall section.



Figure 2. —Composite wall

Remarks:

These equations are for steady state heat transfer through materials with constant properties in all directions.

Inputs must start with the inside convective coefficient and work out in the case of composite cylinders.

Zero is an invalid input for D, k, and h.

Dimensional consistency must be maintained.

Example 1:

A steel pipe with an inside diameter of 4 inches and a thickness of 0.5 inches has a conductivity of 25 Btu/ft-hr-°F. Two inches of asbestos (k=0.1 Btu/hr-ft-°F) enclose the pipe bringing the total diameter to 9 inches. If the inside convective coefficient is 1000 Btu/hr-ft^2-°F and the outside coefficient is 5 Btu/hr-ft^2-°F, what is the overall heat transfer coefficient? What is the heat loss for 100 feet of pipe if ΔT is 115°F?

Example 2:

A wall is composed of 1 foot of brick (k=0.4 Btu/hr-ft-°F), and 1 inch of wood (k=0.12 Btu/hr-ft-°F). The convective coefficient on one side is 23 Btu/hr-ft^2-°F. The convective coefficient of the other side is 5 Btu/hr-ft^2-°F. What is the overall coefficient? What is the heat flux if the temperature difference is 70°F?

Solutions:

1.

```
         CLRG
1000.00  ENT↑
   4.00  ENT↑
  12.00  ÷     (convert units to feet)
         GSB1
  25.00  ENT↑
   5.00  ENT↑
  12.00  ÷
         GSB2
   0.10  ENT↑
   9.00  ENT↑
  12.00  ÷
         GSB2
   5.00  ENT↑
   9.00  ENT↑
  12.00  ÷
         GSB1
   0.98  ***   Btu/hr-ft-°F
 115.00  x
 112.44  ***
 100.00  x
11244.20 ***   Btu/hr
```

2.

```
         CLRG
  23.00  GSB3
   0.40  ENT↑
   1.00  GSB4
   0.12  ENT↑
   1.00  ENT↑
  12.00  ÷
         GSB4
   5.00  GSB3
   0.29  ***   Btu/ft2-hr-°F
  70.00  x
  20.36  ***   Btu/ft2-hr
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1	Key in the program				
2	Initialize		f	REG	
3	For cylinders:				
3a	Enter the convective coefficient and the	h	ENT↑		
	diameter (start from the inside and	D	GSB	1	1/hD or U if outside surface
	work out)				
3b	Enter the conductive coefficient	k	ENT↑		
	and the diameter	D	GSB	2	
3c	Repeat step 3b or go to step 3a for				
	outside surface				
4	For a new case, go to step 2				
5	For walls:				
6a	Enter the convective coefficient (inside	h	GSB	3	1/h or U if final surface
	or outside surface)				
6b	Enter the conductive coefficient and the	k	ENT↑		
	thickness	x	GSB	4	
6c	Repeat step 6b or go to step 6a for final				
	surface				
7	For a new case, go to step 2				

Program Listings

Program	Comments
01 *LBL1	
02 STO7	D h
03 x	
04 Pi	
05 x	
06 *LBL3	
07 1/X	
08 ST+8	
09 RCL8	
10 X=Y?	initial or final entry?
11 R/S	
12 1/X	
13 R/S	**U
14 *LBL2	
15 RCL7	
16 X⇄Y	
17 STO7	
18 ÷	D_{i-1}/D_i
19 LN	
20 X⇄Y	k
21 2	
22 x	
23 Pi	
24 x	
25 ÷	
26 ST-8	
27 R/S	
28 *LBL4	
29 X⇄Y	
30 ÷	x/k
31 ST+8	
32 R/S	

**"Printx" may be inserted before "R/S".

REGISTERS

0	1	2	3	4	5
6	7 D_{i-1}	8 1/U	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

BLACK BODY THERMAL RADIATION

Bodies with finite temperatures emit thermal radiation. The higher the absolute temperature, the more thermal radiation emitted. Bodies which emit the maximum possible amount of energy at every wavelength for a specified temperature are said to be black bodies. While black bodies do not actually exist in nature, many surfaces may be assumed to be black for engineering considerations.



Figure 1.

Figure 1 is a representation of black body thermal emission as a function of wavelength. Note that as temperature increases the area under the curves (total emissive power $E_{b(0-\infty)}$) increases. Also note that the wavelength of maximum emissive power λ_{max} shifts to the left as temperature increases.

This program can be used to calculate the wavelength of maximum emissive power for a given temperature, the temperature corresponding to a particular wavelength of maximum emissive power, the total emissive power for all wavelengths and the emissive power at a particular wavelength. It can also be used to calculate the emissive power from zero to an arbitrary wavelength, the emissive power between two wavelengths or the total emissive power.

EQUATIONS:

$$\lambda_{max} T_{\lambda_{max}} = c_3$$

$$E_{b(0-\infty)} = \sigma T^4$$

$$E_{b\lambda} = \frac{2\pi c_1}{\lambda^5 (e^{c_2/\lambda T} - 1)}$$

$$E_{b(0-\lambda)} = \int_0^{\lambda} E_{b\lambda}\, d\lambda$$

$$= 2\pi c_1 \sum_{k=1}^{\infty} -T/kc_2\ e^{-\frac{kc_2}{T\lambda}} \left[\left(\frac{1}{\lambda}\right)^3 + \frac{3T}{\lambda^2 kc_2} + \frac{6}{\lambda}\left(\frac{T}{kc_2}\right)^2 + 6\left(\frac{T}{kc_2}\right)^3 \right]$$

$$E_{b(\lambda_1-\lambda_2)} = E_{b(0-\lambda_2)} - E_{b(0-\lambda_1)}$$

where

λ_{max} is the wavelength of maximum emissivity in microns;

T is the absolute temperature in °R or K;

$E_{b(0-\infty)}$ is the total emissive power in Btu/hr-ft^2 or Watts/cm^2;

$E_{b\lambda}$ is the emissive power at λ in Btu/hr-ft^2-μm or Watts/cm^2-μm;

$E_{b(0-\lambda)}$ is the emissive power for wavelengths less than λ in Btu/hr-ft^2 or Watts/cm^2;

$E_{b(\lambda_1-\lambda_2)}$ is the emissive power for wavelengths between λ_1 and λ_2 in Btu/hr-ft^2 or Watts/cm^2.

$c_1 = 1.8887982 \times 10^7$ Btu-μm^4/hr-ft^2

$= 5.9544 \times 10^3$ Wμm^4/cm^2

$c_2 = 2.58984 \times 10^4$ μm-°R =

1.4388×10^4 μm-K

$c_3 = 5.216 \times 10^3$ μm-°R =

2.8978×10^3 μm-K

$\sigma = 1.71312 \times 10^{-9}$ Btu/hr-ft^2-°R^4=

5.6693×10^{-12} W/cm^2-K^4

$\sigma_{exp} = 1.731 \times 10^{-9}$ Btu/hr-ft^2-°R^4

$= 5.729 \times 10^{-12}$ W/cm^2-K^4

REMARKS:

A minute or more may be required to obtain $E_{b(0-\lambda)}$ or $E_{b(\lambda_1-\lambda_2)}$ since the integration is numerical.

Sources differ on values for constants. This could yield small discrepancies between published tables and outputs.

REFERENCE:

Robert Siegel and John R. Howell, Thermal Radiation Heat Transfer, Vol. 1, National Aeronautics and Space Administration, 1968.

EXAMPLE 1:

What percentage of the radiant output of a lamp is in the visible range (0.4 to 0.7 microns) if the filament of the lamp is assumed to be a black body at 2400 K?

EXAMPLE 2:

If the human eye was designed to work most efficiently in sunlight and the visible spectrum runs from about 0.4 to 0.7 microns, what is the sun's temperature in degrees Rankine? Assume that the sun is a black body. Using the temperature calculated, find the fraction of the sun's total emissive power which falls in the visible range. Find the percentage of the sun's radiation which has a wavelength less than 0.4 microns.

SOLUTIONS:

1.

```
   5954.40 ST01  }
  14388.00 ST02  }  S.I. constants
   2897.80 ST03  }
5.6693-12  ST04  }
   2400.00 ST05
      0.40 ST06
      0.70 ST07
           GSB4
      4.97  ***
           GSB2   Eb (0 to ∞)
              ÷
    100.00    ×
      2.64  ***   (%)
```

2.

```
18887982.00 ST01   ⎫
   25898.40 ST02   ⎬ English constants
    5216.00 ST03   ⎪
1.71312-09 ST04    ⎭
       0.40 ENT↑
       0.70   +
       2.00   ÷
       0.55  ***     mean value
            RCL3
              ÷
             1/X
    9483.64  ***     T, (°R)
            ST05
       0.40 ST06
       0.70 ST07
            GSB4
 4670556.56  ***     Eb(.4 to .7)
            GSB2
13857578.83  ***     Eb(0 to ∞)
              ÷
     100.00   x
      33.70  ***     (%)
       0.40 ST06
            GSB1
 1168606.94  ***     Eb (0 to .4)
            GSB2
              ÷
     100.00   x
       8.43  ***     (%)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1	Key in the program				
2	Store constants:				
2a	English units -	18887982	STO	1	
	(Btu, μm, ft, °R)	25898.4	STO	2	
		5216	STO	3	
	or	$.171312 \times 10^{-8}$	STO	4	
2b	SI units -	5954.4	STO	1	
	(W, μm, cm, °K)	14388	STO	2	
		2897.8	STO	3	
		5.6693×10^{-12}	STO	4	
3	For experimental Stefan-Boltzmann constant	1.0105	STO	x	
	instead of theoretical constant		4		
4	To calculate $\lambda_{max} = f(T)$		RCL	3	
		T(°Absol.)	÷		λ_{max}(μm)
5	To calculate $T = f(\lambda)$ for which λ is		RCL	3	
	maximum	λ(μm)	÷		T (°Absol.)
6	To calculate total emissive power	T*	STO	5	
			GSB	2	$E_b(0 \text{ to } \infty)$
7	To calculate emissive power at λ	T*	STO	5	
		λ	STO	6	
			GSB	3	$E_b(\lambda)$
8	To calculate emissive power from 0 to λ_1	T*	STO	5	
		λ_1	STO	6	
			GSB	1	$E_b(0 \text{ to } \lambda_1)$
9	To calculate emissive power from λ_1 to λ_2	T*	STO	5	
		λ_1	STO	6	
	*any value of T stored previously is still	λ_2	STO	7	
	stored and need not be input again		GSB	4	$E_b(\lambda_1 \text{ to } \lambda_2)$

Program Listings

Step	Key	Comment	Step	Key	Comment
01	*LBL1		50	X≤Y?	$\Delta \geq .001\%$
02	GSB9		51	GTO0	yes, increment k
03	R/S	***$E_b(0 \text{ to } \lambda_1)$	52	RCL9	
04	*LBL9		53	2	
05	0		54	x	
06	STO9		55	Pi	
07	STO8		56	x	
08	*LBL0		57	RCL1	
09	RCL2		58	x	
10	RCL5		59	RTN	$E_b(0 \text{ to } \lambda)$
11	÷		60	*LBL2	
12	ST-8	$-k\ c_2/T$	61	RCL5	
13	3		62	4	
14	RCL8		63	Y^x	
15	÷		64	RCL4	
16	RCL6		65	x	
17	X^2		66	R/S	**$E_b(0 \text{ to } \infty)$
18	÷		67	*LBL3	
19	LSTX	λ^2	68	RCL1	
20	RCL6		69	2	
21	x		70	x	
22	1/X		71	Pi	
23	-		72	x	
24	6		73	RCL6	
25	RCL6		74	5	
26	÷		75	Y^x	
27	RCL8		76	÷	
28	X^2		77	RCL2	
29	÷		78	RCL6	
30	-		79	÷	
31	6		80	RCL5	
32	RCL8		81	÷	
33	3		82	e^x	
34	Y^x		83	1	
35	÷		84	-	
36	+		85	÷	
37	RCL8		86	R/S	**$E_{b\lambda}$
38	RCL6		87	*LBL4	
39	÷		88	GSB9	
40	e^x		89	ST.0	$E_b(0 \text{ to } \lambda_1)$
41	x		90	RCL7	
42	RCL8		91	STO6	λ_2
43	÷		92	GSB9	
44	ST+9	Δ	93	RC.0	
45	RCL9		94	-	
46	÷		95	R/S	**$E_b(\lambda_1 \text{ to } \lambda_2)$
47	EEX				
48	CHS				
49	5				

**"Printx" may be inserted before "R/S"

REGISTERS

0	1 c_1	2 c_2	3 c_3	4 σ	5 T
6 λ	7 λ_2	8 $-Kc_2/T$	9 sum	.0 used	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

CONSERVATION OF ENERGY

This program converts kinetic energy, potential energy, and pressure-volume work to energy. Energy is stored as a running total which may at any time be converted to an equivalent velocity, height, pressure, or energy per unit mass. The program is useful in fluid flow problems where velocity, elevation and pressure change along the path of flow.

EQUATIONS:

$$\frac{v_1^2}{2} + gz_1 + \frac{P_1}{\rho} + \frac{E_1}{\dot{m}} =$$

$$\frac{v_2^2}{2} + gz_2 + \frac{P_2}{\rho} + \frac{E_2}{\dot{m}}$$

where:

- v is the fluid velocity;
- z is the height above a reference datum;
- P is the pressure;
- E is an energy term which could represent inputs of work or friction loses (negative value);
- g is the acceleration of gravity;
- ρ is the fluid density;
- $\dot{m}$ is the mass flow rate (assumed to be unity);

subscripts 1 and 2 refer to upstream and downstream values respectively.

NOTES:

Downstream values should be input as negatives. However, when an output is called for, the calculator displays the relative value with no regard to upstream or downstream location.

An error will result when the total energy sum stored in register 8 is negative and an attempt is made to calculate velocity.

EXAMPLE 1:

A water tower is 100 feet high. What is the zero flow rate pressure at the base? The density of water is 62.4 lb/ft^3.

If water is flowing out of the tower at a velocity of 10 ft/sec, what is the static pressure?

What is the maximum frictionless flow velocity which could be achieved with the 100 foot tower?

If 10000 pounds of water are pumped to the top of the tower every hour, at a velocity of 20 ft/sec, with a frictional pressure drop of 2 psi, how much power is needed at the pump?

EXAMPLE 2:

An incompressible fluid (ρ = 735 kg/m) flows through the converging passage of Figure 1. At point 1 the velocity is 3 m/s and at point 2 the velocity is 15 m/s. The elevation difference between points 1 and 2 is 3.7 meters. Assuming frictionless flow, what is the static pressure difference between points 1 and 2?



Figure 1.

EXAMPLE 3:

A reservoir's level is 25 meters above the discharge pond. Assuming 85% power generation efficiency, how much power can be generated with a flow rate of 20 m^3/s?

$\rho = 1000\ kg/m^3$

SOLUTIONS:

(1)

```
25033.407 STO5
    32.17 STO6
  4632.48 STO7
    62.40 GSB1
   100.00 GSB3
          GSB8
    43.33  ***  (psig)
   -10.00 GSB2
          GSB8
    42.66  ***  (psig)
    62.40 GSB1
   100.00 GSB3
          GSB6
    80.21  ***  (ft/sec)
    62.40 GSB1
    20.00 GSB2
     2.00 GSB4
   100.00 GSB3
          GSB9
     0.14  ***  (BTU/lb)
 10000.00   x
  1424.29  ***  (BTU/hr)
```

(2)

```
     1.00 STO5
          STO7
  9.80665 STO6
   735.00 GSB1
     3.00 GSB2
     3.70 GSB3
   -15.00 GSB2
          GSB8
-52710.82  ***  (Nt/m²)
```

(3)

```
   1000.00 GSB1
     25.00 GSB3
           GSB9
    245.17  ***  (joule/kg)
      0.85   x
    208.39  ***  (joule/kg)
     20.00 ENT↑
   1000.00   x   (kg/s)
             x
4167826.25  ***  (watts)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	For English units:	25033.407	STO	5	
		32.17	STO	6	
		4632.48	STO	7	1.00
	For S.I. units:	9.80665	STO	6	
3.	Enter fluid density	ρ	GSB	1	0.00
4.	Enter the following (negative values are				
	downstream values):				
	Fluid velocity	v	GSB	2	
	Height from reference datum	z	GSB	3	
	Pressure	P	GSB	4	
	Energy input	E	GSB	5	
5.	Repeat step 4 for all input values				
6.	Calculate the unknown:				
	Fluid velocity		GSB	6	v
	Height from reference datum		GSB	7	z
	Pressure		GSB	8	P
	Energy		GSB	9	E
7.	For another case, go to step 3, or clear				
	register 8 and go to step 4	0	STO	8	

Program Listings

01	*LBL1	
02	STO4	ρ
03	0	
04	STO8	Clear Σ E
05	R/S	
06	*LBL2	
07	ENT↑	
08	ABS	
09	x	
10	2	
11	÷	$\pm\ v^2/_2$
12	GTO5	
13	*LBL3	
14	RCL6	
15	x	
16	GTO5	gz
17	*LBL4	
18	RCL7	
19	x	
20	RCL4	
21	÷	$^{\rho}/\rho$
22	*LBL5	E
23	ST+8	
24	R/S	
25	*LBL6	
26	RCL8	
27	2	
28	x	
29	√X	
30	R/S	*** v
31	*LBL7	
32	RCL8	
33	RCL6	
34	÷	
35	R/S	*** z
36	*LBL8	
37	RCL8	
38	RCL7	
39	÷	
40	RCL4	
41	x	
42	R/S	*** P
43	*LBL9	
44	RCL8	
45	RCL5	
46	÷	
47	R/S	*** E

*** "Printx" may be inserted.

REGISTERS

0	1	2	3	4 ρ	5 Used
6 g	7 Used	8 Σ E	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

NOTES

NOTES

NOTES

In the Hewlett-Packard tradition of supporting HP programmable calculators with quality software, the following titles have been carefully selected to offer useful solutions to many of the most often encountered problems in your field of interest. These ready-made programs are provided with convenient instructions that will allow flexibility of use and efficient operation. We hope that these Solutions books will save your valuable time. They provide you with a tool that will multiply the power of your HP-19C or HP-29C many times over in the months or years ahead.

Mathematics Solutions
Statistics Solutions
Financial Solutions
Electrical Engineering Solutions
Surveying Solutions
Games
Navigational Solutions
Civil Engineering Solutions
Mechanical Engineering Solutions
Student Engineering Solutions

HEWLETT PACKARD

Reorder No. 00029-14009 Printed in U.S.A. 00029-90013